

## Power Series solutions in Maple

The relevant Maple file is a separate document, and the output to the Maple document is also available as a separate PDF.

A power series solution to an ODE may be obtained in one of two different ways:

- Using `powseries` package and `powsolve`
  - This gives a *procedure* from which a truncated series of any order can be produced using `tpsform`
  - Power series is always based at 0.
  - You can extract the recurrence relation directly.
  - Cannot use non-polynomial coefficients
- Using `dsolve` with a `'series'` option
  - Series is based at initial condition.
  - Can use non-polynomial coefficients, non-polynomial forcing.
  - Cannot extract the recurrence relation directly.

In both cases, you can transform a *series* into a truncated polynomial (for plotting, etc) by using `convert`

## Examples using powseries

1. (See PS1.mws) Solve, using a power series:

$$y'' + xy' + 2y = 0, \quad y(0) = a_0, y'(0) = a_1$$

```
with(powseries):
deq:=diff(y(x), x$2) + x*diff(y(x), x) + 2*y(x) = 0;
inits:=y(0)=a[0],D(y)(0)=a[1];
IVP:={deqn,inits};

#The following computes the series solution
#  f is a procedure, F is the series

f:=powsolve(IVP);
F:=tpsform(f,x,12);

#We can extract the recursion:
```

```
f(_k);
```

```
#or, more succinctly:
```

```
recursion_relation:=a(n)=subs(_k=n,f(_k));
```

2. (See PS2.mws) Let

$$y'' - (3x - 2)y' - 2y = 0, \quad y(0) = 0, y'(0) = 2$$

- (a) Find the general power series solution using Maple.
- (b) Find the recurrence relation.
- (c) Compute the order 7 and order 13 approximations.
- (d) Graphically compare the two previous solutions, together with Maple's default solution.
- (e) Is there an  $x$  value for which the solutions suddenly stop matching? See if you can get a good approximation.

```
unassign('y');
```

```
eqn:=diff(y(x),x$2)-(3*x-2)*diff(y(x),x)-2*y(x)=0;  
inits:=y(0)=0,D(y)(0)=2;  
IVP:={eqn,inits};
```

```
with(powseries):  
f:=powsolve(IVP);  
recursion_relation:=a(n)=subs(_k=n,f(_k));
```

```
f6:=tpsform(f,x,7);  
f12:=tpsform(f,x,13);  
F6:=convert(f6,polynomial,x);  
F12:=convert(f12,polynomial,x);
```

```
g:=dsolve(IVP,y(x)); #This will give Maple's default solution  
G:=rhs(g);  
plot({G,F6,F12},x=-1..1,y=-2..2,numpoints=150);
```

## Examples using dsolve

1. (See DS1.mws) Solve, using a power series to order 12:

$$y'' + xy' + 2y = \sin(x), \quad y(1) = a_0, y'(1) = a_1$$

```

unassign('y');
Order:=12;

deq:=diff(y(x), x$2) + x*diff(y(x), x) + 2*y(x) = sin(x);
inits:=y(1)=a[0],D(y)(1)=a[1];
IVP:={deq,inits};

F:=dsolve(IVP,y(x),'series');

# Here is how you convert this into a regular polynomial
# (useful for plotting, if we had numbers in the initial
# conditions).
convert(rhs(F),polynom,x);

```

2. (See Airy1.mws) Construct a picture like Figure 5.2.4, pg. 245 in Boyce and DiPrima. That is, construct the partial power series solutions to Airy's equation,

$$y'' - xy = 0, \quad y(0) = 0, y'(0) = 1$$

We follow the same basic procedures as before. In this case, we could use `powsolve`, but we'll go ahead and use `dsolve`

```

unassign('y');

deq:=diff(y(x),x$2)-x*y(x)=0;
inits:=y(0)=0,D(y)(0)=1;
IVP:={deqn,inits};

Order:=5; #Make this one bigger than the poly degree
f4:=dsolve(IVP,y(x),'series');
F4:=convert(rhs(f4),polynom,x);

#Now repeat those last two lines, changing the order
# to create F4, F10, F16, F22

#And plot them:
plot({F4,F10,F16,F22},x=-10..2,y=-3..3);

```